

Powerful SAS® Output Delivery with ODS Excel

LeRoy Bessler PhD
Bessler Consulting and Research, Mequon, WI, USA
Le_Roy_Bessler@wi.rr.com

Abstract

A common destination for results prepared with SAS® is often an Excel workbook. Everyone already has Excel and knows how to use it, to reformat or further explore their results however they wish. ODS Excel enables a SAS programmer to create highly formatted reports, tabular or graphic, or a combination of both, that can be opened and used with Excel. You can turn on customization/formatting features in SAS that would be possible manually inside Excel, to deliver an already finished product to the viewer of the report. The ODS Excel capability does not require Excel to be installed on the machine that creates ODS Excel output. You can use ODS EXCEL running SAS on MVS, UNIX, Linux, or Windows. This paper assumes no prior knowledge of the topic. ODS Excel output requires Microsoft Excel 2010 or later.

Introduction

In the winter of 2012-2013, I was working on an application where, besides graphs, I needed to provide lots of Excel spreadsheets, linked backwards and forwards with the graphs, and to let summary spreadsheets have links to detail spreadsheets for each summary row. And I wanted a lot of control over spreadsheet features and format. My frustration was that there was no single SAS solution to address ALL of my spreadsheet function and feature needs. After this experience, I made two successive (published) forays into comparing the capabilities of the various SAS-provided tools. In the **Options Available Prior to the ODS Excel Destination** section below, I summarize my conclusions.

My first in-depth adventure, ten years earlier, with using SAS to create highly formatted reports that can be opened with Excel was DDE (Dynamic Data Exchange). See Reference 1 (the latest update to my original 2003 paper), which mentions a DDE toolkit for which I still get requests frequently sixteen years later. Unless you need to create a pivot table, for which the best resource is Reference 2, I suspect that DDE will remain a popular solution for some SAS users. Your SAS program runs as a client of an Excel session that your program starts to serve as its Excel server. It's only clumsy to deal with when using Enterprise Guide and a remote SAS server. If something goes wrong, you end up with a hung SAS process and a hung Excel process that are out of your direct control. Reference 3 provides tools to deal with the SAS process, and analogous tools can be created to deal with the hung Excel process.

The main content of this paper is the section **ODS Excel Examples**, which covers all non-printing features in ODS Excel available as of August 2019 in dozens of examples. Before getting into the examples, there is a comparison of Options Available Prior to the ODS Excel Destination, as well as a section on Set-Up and Common Code for All of the Examples.

Acknowledgements

Essential to my initial project on this topic was assistance from Wayne Hester, Nancy Goodling, Chevell Parker, Scott Huntley, Dan O'Connor, and Amy Peters at SAS Institute. Thereafter, I got help from Jane Eslinger and Cynthia Zender at SAS Technical Support, and more recently from Chevell Parker and Martin Mincey of SAS Technical Support. Any errors or imperfections in this paper are my responsibility.

Options Available Prior to the ODS Excel Destination

Some users might have option interests other than those listed below, but it is my judgement that these are the ones most commonly desired.

Feature	<u>TableEditor</u> *	<u>ExcelXP</u>	<u>MSOffice2K_x</u>	<u>HTML</u>	<u>DDE</u> **
Pivot Table	Yes	No	No	No	requires an Excel macro
AutoFit Columns	By Default***	No	By Default***	By Default	Yes
Filters	Yes, but All Columns only	Yes	Yes, but All Columns only	No	Yes
Freeze Panes	Row 1 only	Yes	Yes	No	Yes
Side-By-Side Elements in <u>WorkSheet</u>	Yes	No	Yes	No	Yes
Multi-Sheet <u>WorkBook</u>	Yes	Yes, Easy	Yes	No	Yes
Graphs	Yes	No	Yes	Yes	Yes
Titles	Not with other features	Yes	Yes	Yes	Yes

*Javascript must be enabled

**SAS must be running on Windows

***Headings of columns autofit by default can be overlaid by filter buttons

Note: The comparison above omits MSOffice2K. It was not evaluated by me because it has fewer capabilities than MSOffice2K_x.

Set-Up and Common Code for All of the Examples

```
/* Common SetUp
   with folder assignments specific to MY computer */
```

```
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
```

```
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
```

NOTE: The ImgPath folder is used in this paper only for storage of permanent image files. Some examples create images that are stored in C:\temp. However, the %LET statement appears (unnecessarily in most cases) in all example code.

```
/* DemoX Short Description */
%let N = X; /* Demo Step */
%let ZoomPct = xxx; /* adjusted to fill screen for demo */
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
/* NOTE: All footnotes are optional. If omitting FOOTNOTE3,
   then omit %INCLUDE and invocation of RunDayDateTime macro. */
```

```
/* Common Framing: */
```

```
ods results off; ods _all_ close;
ods excel file="&Path\Demo&N..xlsx"
  options( . . . );
ods excel close; ods results on;
```

```
/* All of the XLSX filenames created are actually of the form
   "DemoXX Text Description"
   but FOOTNOTE2 displays a truncated filename of the form DemoXX */
```

Preliminary Comments

ODS Excel features related to printing are NOT demonstrated in this paper.

Any graphs used in this paper were not necessarily created with regard to design principles that I strenuously advocate elsewhere. Their purpose and presence here is solely to demonstrate how to deliver graphs using ODS Excel, not how best to design and create them.

In most cases, ALL of the code used is shown here. However, a zip file of all of the code for every example and for all of the macros can be requested from Le_Roy_Bessler@wi.rr.com .

Where possible, key distinguishing elements of the code are highlighted with **blue** or **red** and use bold **SAS Monospace Bold** font. In some cases, there is too much code that is essential and distinguishing for the example.

To make the nonblank parts of a worksheet maximally readable in screen capture images used in this paper, the Excel Zoom feature was used. In most of the examples, the Zoom percent is identified in the worksheet TITLE line. Excel 2010 was used to view the outputs created for this paper. If you use a different version of Excel, the results might look different. ODS Excel REQUIRES use of Excel 2010 or a later version.

NOTE: If, when trying to create a graph in an ODS Excel worksheet, you get this message:

WARNING: WIDTH exceeds available space for EXCEL destination

the current limit that is cited might actually be due to some prior processing during the same SAS session. It was my experience that the REAL limit is 8 inches.

Special Cases

Four examples (Demo25, Demo26, Demo28, Demo29) rely on use of the Excel_Enhance macro, not just ODS Excel. These examples rely on the macro's insert_image function (which also entails the use of three other macros). Excel_Enhance allows you to place an image file with its upper left corner in any cell of a worksheet. You can place MULTIPLE images.

Without the Excel_Enhance macro, the ODS Excel Start_At option can locate a table anywhere on the worksheet, but you can do this for ONLY ONE table and only if it is the first output created by the code being run to create the workbook.

NOTE: A link to the downloadable Excel_Enhance macro used for this paper is in Appendix B. The version used here was downloaded on 17 August 2019. If you have any problems with the link (e.g., it no longer exists), please contact SAS Technical Support.

ODS Excel Examples

By Category and Purpose

In Suggested Reading Order, NOT in sequence of appearance in this paper

Basic Features

Demo01 Starting Example
Demo02 Custom Sheet Name
Demo03 Preventing Title Line Wrap
Demo04 TITLE2 As A HyperLink
Demo05 Freezing Headers and Row Headers
Demo06 AutoFilters
Demo07 Verify Maximum Sheet-Name Length
Demo08 Controlling the Upper Left Corner of the First Output on a WorkSheet
Demo09 Hiding Rows and Columns
Demo10 Controlling WorkSheet Tab Color
Demo11 Creating an Empty WorkSheet
Demo12 Protecting a WorkSheet from Changes

Titles and Footnotes for Graphs Inside Versus Outside of Image Area

Demo18 Effect of NOGTITLE NOGFOOTNOTE Options
Demo19 Effect of GTITLE GFOOTNOTE Options

Organizing and Delivering Multiple Outputs

Demo20 PROC UNIVARIATE Outputs in Separate WorkSheets
Demo21 PROC UNIVARIATE Outputs in the Same WorkSheet
Demo22 Two Tables Stacked
Demo23 Two Tables Side By Side Using PROC PRINT
Demo24 Two Tables Side By Side Using PROC REPORT
Demo25 Four Tables Side By Side Using Excel_Enhance
Demo26 Four Tables in a Two By Two Array Using Excel_Enhance
Demo27 Two Plots Stacked
Demo28 Two Plots Side By Side Using Excel_Enhance
Demo29 Two Plots & Two Tables Using Excel_Enhance
Demo30 Bar Chart & Table in Separate WorkSheets
Demo31 Bar Chart Over Table in the Same WorkSheet
Demo32 Table Over Bar Chart in the Same WorkSheet
Demo33 Hyperlinks and Non-Graphic Images Over and Under Table

Index or Table of Contents

Demo34 ODS Excel Index
Demo35 ODS Excel Index With All WorkSheets InterLinked
Demo36 ODS Excel Index with SAS ByGroup Processing
Demo37 ByGroup Processing With Excel Index and LinkBack
Demo38 ODS Excel Table of Contents With All WorkSheets InterLinked

Formatting Numeric Data

Demo13 TAGATTR for Static Microsoft Numeric Formats
Demo14 TAGATTR for a Dynamic Microsoft Numeric Format

Controlling Widths and Heights

Demo15 Controlling Title and Footnote Width and Row Heights
Demo16 Controlling Column Widths
Demo17 Controlling Column Widths When Using FixedWidth Font

Demo01 Starting Example

Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

14 Source Data: SASHELP.SHOES
 16 Code: C:\MWSUG_2019\ODS_Excel\code\Demo01.sas
 17 Run on: Wednesday, 31 July 2019 at 8:58:56 PM

```

/* Simple Example, Default Sheet Name, Title Line Wrap, File "Properties" */
%let N = 01; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 240; /* want to nearly fill the Excel window */
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo01 Starting Example.xlsx"
  title='Demo01 Starting Example'
  author='LeRoy Bessler PhD'
  keywords='ODS Excel'
  comments='Used SAS 9.4 TS1M06 and Excel 2010'
  status='Demo for MWSUG 2019'
  options(embedded_titles='yes' embedded_footnotes='yes'
    zoom="&ZoomPct" );
title1 justify=left bold
  "Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom "
  color=red "See Default Sheet Name "
  color=blue "(maximum 24 characters)";
options obs=&OBScount;
proc print data=sashelp.shoes noobs;
id region subsidiary product;
run;
options obs=max;

```

```

title; footnote;
ods excel close; ods results on;
/* The ODS Excel Statement options TITLE, KEYWORDS (Tags in Excel), COMMENTS,
STATUS, CATEGORY, and AUTHOR are optional. Their content appears in the Excel
File Properties accessed by clicking File and then Info. Below is a cropped
and enlarged screen capture of the Properties for this Demo01 workbook. */

```

Demo01 Starting Example.xlsx - Microsoft Excel

Region	Subcategory	Product	Shirts	Suits	Sweaters	Returns
Africa	Adults/Alaska	Boat	12	\$22,701	\$191,021	\$788
Africa	Adults/Alaska	Men's Casual	4	\$17,343	\$110,038	\$2,284
Africa	Adults/Alaska	Men's Dress	7	\$76,790	\$136,213	\$2,433
Africa	Adults/Alaska	Swimsuit	10	\$52,816	\$234,254	\$1,861
Africa	Adults/Alaska	Slipper	14	\$55,941	\$210,755	\$1,771
Africa	Adults/Alaska	Sport Shoes	4	\$1,646	\$10,334	\$10
Africa	Adults/Alaska	Women's Casual	2	\$21,341	\$68,641	\$340
Africa	Adults/Alaska	Women's Dress	10	\$100,842	\$211,017	\$2,102
Africa	Alps	Boat	21	\$11,281	\$73,737	\$110
Africa	Alps	Men's Casual	4	\$13,236	\$100,082	\$2,221

Properties ▾

Size 6.55KB

Title Demo01 Starting Example

Tags ODS Excel

Comments Used SAS 9.4 TS1M06 and Excel 2...

Template

Status Demo for MWSUG 2019

Categories ODS Excel Example

Subject Specify the subject

Hyperlink Base Add text

Company Specify the company

Related Dates

Last Modified Today, 4:46 PM

Created Today, 4:46 PM

Last Printed Never

Related People

Manager Specify the manager

Author LeRoy Bessler PhD
Add an author

Last Modified By LeRoy Bessler PhD

Related Documents

Open File Location

[Show Fewer Properties](#)

Demo02 Custom Sheet Name

Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

Source Data: SASHELP.SHOES
Code: C:\MWSUG_2019\ODS_Excel\code\Demo02.sas
Run on: Wednesday, 31 July 2019 at 8:59:00 PM

```

/* Custom Sheet Name, But Still Title Line Wrap */
%let N = 02; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 230;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo02 Custom Sheet Name.xlsx"
      options( embedded_titles='yes' embedded_footnotes='yes'
              zoom="&ZoomPct" sheet_name='Custom Sheet Name' );
title1 justify=left bold
      "Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom "
      color=red
      "Custom Sheet Name is 'Custom Sheet Name' "
      color=blue
      "Title still wraps in the merged area above data columns.";
options obs=&OBScount;
proc print data=sashelp.shoes noobs;
id region subsidiary product;
run;
options obs=max;
title; footnote;
ods excel close; ods results on;

```


Demo03 Preventing Title Line Wrap

Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

Source Data: SASHELP.SHOES
 Code: C:\MWSUG_2019\ODS_Excel\code\Demo03.sas
 Run on: Wednesday, 31 July 2019 at 8:59:02 PM

```

/* title_footnote_nobreak='yes' cures Title Line Wrap */
%let N = 03; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 240;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo03 Preventing Title Line Wrap.xlsx"
  options( embedded_titles='yes' embedded_footnotes='yes'
           zoom="&ZoomPct" sheet_name='Title Wrap Cured'
           title_footnote_nobreak='yes' );
title1 justify=left bold
  "Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom "
  color=red
  "Custom Sheet Name is 'Title Wrap Cured'";
title2 justify=left bold color=blue
  "option title_footnote_nobreak='yes' extends title space beyond the
  columns used for data";
options obs=&OBScount;
proc print data=sashelp.shoes noobs;
id region subsidiary product;
run;
options obs=max;
title; footnote;
ods excel close; ods results on;

```

Demo04 TITLE2 as a HyperLink (not really a capability related to ODS Excel)

The screenshot shows an Excel spreadsheet with the following content:

Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

Below the table, the following text is displayed:

Source Data: SASHELP.SHOES
 Code: C:\MWSUG_2019\ODS_Excel\code\Demo04.sas
 Run on: Wednesday, 31 July 2019 at 8:59:04 PM

The title of the spreadsheet is "Demo04 - 10 Obs - Shoe Sales - 240% Zoom making TITLE2 a hyperlink". The second row contains a blue underlined hyperlink: "Go to Wisconsin Illinois SAS Users Web Site Home Page".

```

/* (Could link to another workbook, */
/* a web page, or other filetype, etc.) */
%let N = 04; /* Demo Step */%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath  = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 240;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo04 TITLE2 As a HyperLink.xlsx"
  options(embedded_titles='yes' embedded_footnotes='yes'
    zoom="&ZoomPct" sheet_name='Title2 Is Hyperlink');
title1 justify=left bold
  "Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom "
  color=red "making TITLE2 a hyperlink";
title2 justify=left bold color=blue underlin=1
link='http://www.wiilsu.org'
'Go to Wisconsin Illinois SAS Users Web Site Home Page';
options obs=&OBScount;
proc print data=sashelp.shoes noobs;
id region subsidiary product;
run;
options obs=max;
title; footnote;
ods excel close; ods results on;

```

Demo05 Freezing Headers and Row Headers

Demo05 - Shoe Sales - 10 Obs - 260% Zoom freezing column headers & row headers						
Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

Source Data: SASHELP.SHOES
Code: C:\MWSUG_2019\ODS_Excel\code\Demo05.sas
Run on: Thursday, 1 August 2019 at 2:06:26 PM

```

%let N = 05; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 260;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods all close;
ods excel file="&Path\Demo05 Freezing Headers and Row Headers.xlsx"
  options( embedded_titles='yes' embedded_footnotes='yes'
    zoom="&ZoomPct"
    sheet_name='Freeze Headers + RowHeaders'
    /* & in sheet_name would get converted to - */
    title_footnote_nobreak='yes'
    frozen_headers='3' frozen_rowheaders='3' );
title1 justify=left bold
  "Demo&N - Shoe Sales - &OBScount Obs - &ZoomPct.% Zoom "
  color=red "freezing column headers & row headers";
options obs=&OBScount;
proc print data=sashelp.shoes noobs;
id region subsidiary product; run;
options obs=max;
title; footnote;
ods excel close; ods results on;

```

Demo06 AutoFilters

Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

Source Data: SASHELP.SHOES
 Code: C:\MWSUG_2019\ODS_Excel\code\Demo06.sas
 Run on: Thursday, 1 August 2019 at 2:06:34 PM

CloseUp of the Filters:

Region	Subsidiary	Product	Stores
Africa	Addis Ababa	Boot	12

```

%let N = 06; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 260;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo06 AutoFilters.xlsx"
  options( embedded_titles='yes' embedded_footnotes='yes'
    zoom="&ZoomPct" autofilter='1-4'
    sheet_name='AutoFilters Columns A to D'
    title_footnote_nobreak='yes'
    frozen_headers='3' frozen_rowheaders='3' );
title1 justify=left bold
  "Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom "
  color=red "autofilters for Columns A to D";
options obs=&OBScount;
proc print data=sashelp.shoes noobs; id region subsidiary product;run;
options obs=max;
title; footnote;
ods excel close; ods results on;
  
```


Demo07 Verify Maximum Sheet-Name Length (which is 28)

1 Demo07 - 10 Obs - Shoe Sales - 240% Zoom verifying that maximum sheet name length is 28
 2 Convince yourself by appending 9 to the end of the sheet_name in the code used here.

Region	Subsidiary	Product	Storer	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

16 Source Data: SASHELP.SHOES
 17 Code: C:\MWSUG_2019\ODS_Excel\code\Demo07.sas
 18 Run on: Wednesday, 31 July 2019 at 8:59:10 PM

18 Run on: Wednesday, 31 July 2019 at 8:59:10 PM

19

1234567891123456789212345678

Ready

```

/* Demonstrate that sheet name length accommodates 28 characters */
%let N = 07; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 240;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo07 Verify Maximum Sheet-Name Length.xlsx"
  options( embedded_titles='yes' embedded_footnotes='yes'
           zoom="&ZoomPct" autofilter='1-4'
           title_footnote_nobreak='yes'
           frozen_headers='4'
           frozen_rowheaders='3'
           sheet_name='1234567891123456789212345678' );
title1 justify=left bold
  "Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom "

```

```
color=red "verifying that maximum sheet name length is 28";  
title2 justify=left bold color=blue  
"Convince yourself by appending 9 to the end of the sheet_name in the  
code used here.";  
options obs=&OBScount;  
proc print data=sashelp.shoes noobs;  
id region subsidiary product; run;  
options obs=max;  
title; footnote;  
ods excel close; ods results on;
```

Demo08 Controlling the Upper Left Corner of the First Output on a WorkSheet

Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

Source Data: SASHELP.SHOES
Code: C:\MWSUG_2019\ODS_Excel\code\Demo08.sas
Run on: Wednesday, 31 July 2019 at 8:59:12 PM

```

/* Start_At=(3,2) for Col3 Row2 */
/* NOTE 1: Start_At can be used only once in a worksheet. */
/* NOTE 2: In a multi-PROC ODS package, Start_At must appear as an OPTIONS
assignment BEFORE the first PROC step; otherwise it has no effect. */
%let N = 08; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 250;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods all close;
ods excel file="&Path\Demo08 Controlling Upper Left Corner First Output on a
WorkSheet.xlsx"
    options( embedded_titles='yes' embedded_footnotes='yes'
            title_footnote_nobreak='yes'
            frozen_headers='4' frozen_rowheaders='5'
            start_at='3,2' autofilter='1-4' zoom="&ZoomPct"
            sheet_name='Start Content at Col3Row2' );
title1 justify=left bold
"Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom "
    color=red "Start content at Column 3 Row 2";
options obs=&OBScount;
proc print data=sashelp.shoes noobs;
id region subsidiary product; run;
options obs=max;
title; footnote; ods excel close; ods results on;

```


Demo09 Hiding Rows and Columns

Region	Subsidiary	Product	Stores	Sales
Africa	Addis Ababa	Boot	12	\$29,761
Africa	Addis Ababa	Men's Casual	4	\$67,242
Africa	Addis Ababa	Men's Dress	7	\$76,793
Africa	Addis Ababa	Sandal	10	\$62,819
Africa	Addis Ababa	Sport Shoe	4	\$1,690
Africa	Addis Ababa	Women's Casual	2	\$51,541
Africa	Addis Ababa	Women's Dress	12	\$108,942
Africa	Algiers	Boot	21	\$21,297
Africa	Algiers	Men's Casual	4	\$63,206


Source Data: SASHELP.SHOES
Code: C:\MWSUG_2019\ODS_Excel\code\Demo09.sas
Run on: Wednesday, 31 July 2019 at 8:59:14 PM

```

/* Hide row 8 & cols 6-7 */
%let N = 09; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 270;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo09 Hiding Rows and Columns.xlsx"
  options( embedded_titles='yes' embedded_footnotes='yes'
    zoom="&ZoomPct" autofilter='1-4'
    title_footnote_nobreak='yes'
    frozen_headers='3' frozen_rowheaders='3'
    sheet_name='hide row 8 and cols 6-7'
    hidden_rows='8' hidden_columns='6-7' );
title1 justify=left bold
  "Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom "
  color=red "hide row 8 and columns 6-7";
options obs=&OBScount;
proc print data=sashelp.shoes noobs;
id region subsidiary product; run;
options obs=max;
title; footnote;
ods excel close; ods results on;

```

Demo10 Controlling WorkSheet Tab Color



CloseUp of Tab When Girls WorkSheet is Active:

15	Mary	F	15	66.5	112.0
16					
17	Source Data: SASHELP.CLASS				
18	Code: C:\MWSUG_2019\ODS_Excel\code\Demo10.sas				
19	Run on: Thursday, 1 August 2019 at 2:52:56 PM				
20					
21					

Ready

CloseUp of Tab When Boys WorkSheet is Active:

16	William	M	15	66.5	112.0
17					
18	Source Data: SASHELP.CLASS				
19	Code: C:\MWSUG_2019\ODS_Excel\code\Demo10.sas				
20	Run on: Thursday, 1 August 2019 at 2:52:56 PM				

Ready

```

%let N = 10; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 215;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";

title1 justify=left bold
    "Demo&N - "
    color=red
    "WorkBook with Color-Coded WorkSheet Tabs "
    color=black
    "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue
    "At WorkBook Open & After, the Active Sheet tab has only a tinge of
color at its bottom";
title3 justify=left bold color=blue
    "After either Sheet tab has been pressed, the InActive sheet tab is
fully colored";
ods results off; ods _all_ close;
ods excel file="&Path\Demo10 Controlling WorkSheet Tab Color.xlsx"
    options(embedded_titles='yes' embedded_footnotes='yes'
        title_footnote_nobreak='yes'
        sheet_interval='proc' zoom="&ZoomPct");


ods excel options( sheet_name='Girls' tab_color='LightRed' );
title4 justify=left bold
    "Female Students - &ZoomPct.% Zoom";
proc print data=sashelp.class noobs;
where sex eq 'F';
id name;
run;

ods excel options( sheet_name='Boys' tab_color='LightBlue' );
title4 justify=left bold
    "Male Students - &ZoomPct.% Zoom";
proc print data=sashelp.class noobs;
where sex eq 'M';
id name;
run;


title; footnote;
ods excel close; ods results on;

```


Demo11 Creating an Empty WorkSheet



CloseUp of Tab for Inactive Empty WorkSheet:



Unexciting Size-Reduced Image of Screen Print of Empty WorkSheet:



/* Empty worksheet has Girls & Boys companion sheets

*/

```

/* For more than one empty sheet, your sheet_name= is ignored */
/* They will be named Sheet1, Sheet2, etc. */
%let N = 11; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 240;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
title1 justify=left bold
    "Demo&N - "
    color=red
    "Includes a Custom-Named Empty WorkSheet with Tab Color Blue "
    color=black
    "- &ZoomPct.% Zoom";
ods results off; ods _all_ close;
ods excel file="&Path\Demo11 Creating an Empty WorkSheet.xlsx"
    options( embedded_titles='yes' embedded_footnotes='yes'
        title_footnote_nobreak='yes'
        sheet_interval='proc' zoom="&ZoomPct" );
ods excel options(sheet_name='Girls');
title2 justify=left bold "Female Students";
proc print data=sashelp.class noobs;where sex eq 'F'; run;
ods excel options(sheet_name='Boys');
title2 justify=left bold "Male Students";
proc print data=sashelp.class noobs; where sex eq 'M'; run;
ods excel options( blank_sheet='Custom Named Empty Sheet' tab_color='blue' );
title; footnote;
ods excel close; ods results on;

```

Demo12 Protecting a WorkSheet from Changes

Trying to Change the Sales Value for Men's Dress Shoes:

Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Sand	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Slipp	12	\$108,942	\$311,017	\$3,233
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

CloseUp of Pop-Up Message:



The cell or chart that you are trying to change is protected and therefore read-only.

To modify a protected cell or chart, first remove protection using the Unprotect Sheet command (Review tab, Changes group). You may be prompted for a password.

OK

```
/* Protect_Sheet='yes' prevents change to the worksheet */
%let N = 12; /* Demo Step */
%let Path = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 240;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo12 Protecting a WorkSheet from Changes.xlsx"
  options( embedded_titles='yes' embedded_footnotes='yes'
           protect_worksheet='yes'
           zoom="&ZoomPct" sheet_name='protect_worksheet_EQ_Yes'
           title_footnote_nobreak='yes' );
title1 justify=left bold
  "Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom";
title2 justify=left bold color=red
  "protect_worksheet='Yes' - This Sheet CAN NOT Be Changed";
options obs=&OBScount;
```



```

proc print data=sashelp.shoes noobs;
id region subsidiary product;
run;
options obs=max;
ods excel
    options( protect_worksheet='no' /* this is actually the default */
            sheet_name='protect_worksheet_EQ_No' );
title1 justify=left bold
    "Demo&N - &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom";
title2 justify=left bold color=blue
    "protect_worksheet='No' (this is the default) - This Sheet CAN Be
Changed";
options obs=&OBScount;
proc print data=sashelp.shoes noobs;
id region subsidiary product;
run;
options obs=max;
title; footnote;
ods excel close; ods results on;

```


Demo13 TAGATTR for Static Microsoft Numeric Formats

The screenshot shows an Excel spreadsheet with the following data:

Country	Predicted Sales	Actual Sales	Difference
CANADA	\$233,019	\$246,990	\$13,971
GERMANY	\$231,554	\$245,998	\$14,444
U.S.A.	\$241,722	\$237,349	(\$4,373)

Below the table, the following source information is displayed:

- Source Data: SASHELP.SHOES
- Code: C:\MWSUG_2019\ODS_Excel\code\Demo13.sas
- Run on: Wednesday, 31 July 2019 at 8:59:22 PM

```

/* So that Excel formats data just as you, SAS Programmer, want it */
%let N = 13; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 340;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
data work.ToSummary;
set sashelp.prdsale; Difference = actual - predict; run;
proc summary data=work.ToSummary nway;
class country; var predict actual Difference;
output out=ToPrint sum=; run;
ods results off; ods _all_ close;
ods excel file="&Path\Demo13 TAGATTR for Static Microsoft Numeric
Formats.xlsx"
    options( embedded_titles='yes' embedded_footnotes='yes'
            title_footnote_nobreak='yes'
            zoom="&ZoomPct" sheet_name='Using Tag Attributes' );
title1 justify=left bold
    "Demo&N - "
    color=red
    "Actual Vs Predicted Sales By Country "
    color=black
    "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue
    "TAGATTR in ODS style parameter to apply static Microsoft numeric
formats";
proc print data=work.ToPrint noobs label; id country;

```

```
var predict actual /
    style={tagattr='format:$#,##0_};[Red]\($#,##0\)'};
var Difference /
    style={tagattr='format:$#,##0_};[Red]\($#,##0\)'};
run;
title; footnote;
ods excel close; ods results on;
```

Demo14 TAGATTR for a Dynamic Microsoft Numeric Format

Demo14 - Actual Vs Predicted Sales By Country - 290% Zoom

TAGATTR in ODS style parameter to apply dynamic Microsoft numeric format

Any Manual Changes to Actual and/or Predicted
Will Be Reflected in a Revised Difference

Country	Predicted Sales	Actual Sales	Difference
CANADA	\$233,019	\$246,990	\$13,971
GERMANY	\$231,554	\$245,998	\$14,444
U.S.A.	\$241,722	\$237,349	(\$4,373)

Source Data: SASHELP.SHOES
Code: C:\MWSUG_2019\ODS_Excel\code\Demo14.sas
Run on: Wednesday, 31 July 2019 at 8:59:24 PM

After Changing U.S.A. Actual Sales to \$251,722 and Clicking on the Difference Cell:

Demo14 - Actual Vs Predicted Sales By Country - 290% Zoom

TAGATTR in ODS style parameter to apply dynamic Microsoft numeric format

Any Manual Changes to Actual and/or Predicted
Will Be Reflected in a Revised Difference

Country	Predicted Sales	Actual Sales	Difference
CANADA	\$233,019	\$246,990	\$13,971
GERMANY	\$231,554	\$245,998	\$14,444
U.S.A.	\$241,722	\$251,722	\$10,000

Source Data: SASHELP.SHOES
Code: C:\MWSUG_2019\ODS_Excel\code\Demo14.sas
Run on: Wednesday, 31 July 2019 at 8:59:24 PM

```
%let N = 14; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 290;
```

```

footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
data work.ToSummary;
set sashelp.prdsale;
Difference = actual - predict;
run;
proc summary data=work.ToSummary nway;
class country;
var predict actual Difference;
output out=ToPrint sum=;
run;
ods results off; ods _all_ close;
ods excel file="&Path\Demo14 TAGATTR for a Dynamic Microsoft Numeric
Format.xlsx"
    options(embedded_titles='yes' embedded_footnotes='yes'
            title_footnote_nobreak='yes'
            sheet_name='Tag Attributes and Formula' zoom="&ZoomPct");
title1 justify=left bold
    "Demo&N - "
    color=red
    "Actual Vs Predicted Sales By Country "
    color=black
    "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue
    "TAGATTR in ODS style parameter to apply dynamic Microsoft numeric
format";
title3 justify=left bold color=red
    "Any Manual Changes to Actual and/or Predicted"
    justify=left
    "Will Be Reflected in a Revised Difference";
proc print data=work.ToPrint noobs label;
id country;
var predict actual /
    style={tagattr='format:$#,##0_};[Red]\($#,##0\)'};
var Difference /
    style(data)={tagattr='format:$#,##0_};[Red]\($#,##0\)
    formula:RC[-1]-RC[-2]'};
/* style(data) required to avoid corruption of column header
when FORMULA is mistakenly applied to the header for the column */
run;
title; footnote;
ods excel close; ods results on;

```

Demo15 Controlling Title and Footnote Width and Row Heights

Demo15 - First 10 Obs - Shoe Sales - 230% Zoom
 Title & Footnote Width & Row Height Controls
 AutoFilter for Columns A to D

Region	Subsidiary	Product	Stores	Sales	Inventory	Returns
Africa	Addis Ababa	Boot	12	\$29,761	\$191,821	\$769
Africa	Addis Ababa	Men's Casual	4	\$67,242	\$118,036	\$2,284
Africa	Addis Ababa	Men's Dress	7	\$76,793	\$136,273	\$2,433
Africa	Addis Ababa	Sandal	10	\$62,819	\$204,284	\$1,861
Africa	Addis Ababa	Slipper	14	\$68,641	\$279,795	\$1,771
Africa	Addis Ababa	Sport Shoe	4	\$1,690	\$16,634	\$79
Africa	Addis Ababa	Women's Casual	2	\$51,541	\$98,641	\$940
Africa	Addis Ababa	Women's Dress	12	\$108,942	\$311,017	\$3,233
Africa	Algiers	Boot	21	\$21,297	\$73,737	\$710
Africa	Algiers	Men's Casual	4	\$63,206	\$100,982	\$2,221

15 Source Data: SASHELP.SHOES
 16 Code: C:\MWSUG_2019\ODS_Excel\code\Demo15.sas
 17 Run on: Wednesday, 31 July 2019 at 8:59:26 PM

```
%let N = 15; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 230;
%let OBScount = 10;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo15 Controlling Title and Footnote Width and Row
Heights.xlsx"
```

```
options( embedded_titles='yes' embedded_footnotes='yes'
title_footnote_width='4' /* 4 columns */
row_heights='14,14,0,45,14'
autofilter='1-4'
sheet_name='TextWidth + RowHgt Controls'
zoom="&ZoomPct" );
```

/* ROW_HEIGHTS = 'number_list' specifies height of rows in pts.

**The parameters of this option are positional,
but not all values must be specified.**

**A value of 0 means that the height
should be taken from the style.**

1st value = height for table header rows

2nd value = height for table body rows

3rd value = height for BY line rows

4th value = height for title rows

5th value = height for footnote rows

6th value = height for page break

7th value = height for paragraph skip */

```
title1 justify=left bold
```

```
"Demo&N - First &OBScount Obs - Shoe Sales - &ZoomPct.% Zoom "
```

```
color=red
```

```
"Title & Footnote Width & Row Height Controls "
```

```
color=blue
```

```
"AutoFilter for Columns A to D";
```

```
options obs=&OBScount;
```

```
proc print data=sashelp.shoes noobs;
```

```
id region subsidiary product;
```

```
run;
```

```
options obs=max;
```

```
title; footnote;
```

```
ods excel close; ods results on;
```


Demo16 Controlling Column Widths

This choice of widths, after some experimentation, works well.

Knowing the correct width beforehand for characters of varying width is impossible.

The first three columns could be one “character” width unit smaller, but that cramps “Jeffrey” and the column headings “Sex” and “Age”.

	A	B	C	D	E	F	G	H	I	J	K
1	Demo16 - Students Whose Name Starts with 'J' - 270% Zoom										
2	Using absolute_column_width='6,4,4,6,6' with default width unit of "Characters"										
3	The default Style for ODS Excel uses the Arial font which has characters of varying width										
4											
5	Name	Sex	Age	Height	Weight						
6	James	M	12	57.3	83.0						
7	Jane	F	12	59.8	84.5						
8	Janet	F	15	62.5	112.5						
9	Jeffrey	M	13	62.5	84.0						
10	John	M	12	59.0	99.5						
11	Joyce	F	11	51.3	50.5						
12	Judy	F	14	64.3	90.0						
13											
14	Source Data: SASHELP.CLASS										
15	Code: C:\MWSUG_2019\ODS_Excel\code\Demo16.sas										
16	Run on: Thursday, 1 August 2019 at 4:49:30 PM										

```
%let N = 16; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 270;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo16 Controlling Column Widths.xlsx"
  options( embedded_titles='yes' embedded_footnotes='yes'
    title_footnote_nobreak='yes' zoom="&ZoomPct"
    sheet_interval='proc' );
title1 justify=left bold
  "Demo&N - Students Whose Name Starts with 'J' - &ZoomPct.% Zoom";
ods excel options( sheet_name='Widths 6,4,4,6,6'
  absolute_column_width='6,4,4,6,6' );
title2 justify=left bold color=blue
  "Using absolute_column_width='6,4,4,6,6'"
  ' with default width unit of "Characters"';
title3 justify=left bold color=blue
  'The default Style for ODS Excel uses the Arial font which has
characters of varying width';
proc print data=sashelp.class noobs;
where Name =: 'J';
run;
```



```

ods excel options( sheet_name='All Width 10'
                  absolute_column_width='10' );
title2 justify=left bold color=blue
      "Using absolute_column_width='10'"
      ' with default width unit of "Characters"';
title3 justify=left bold color=blue
      'The default Style for ODS Excel uses the Arial font which has
characters of varying width';
proc print data=sashelp.class noobs;
where Name =: 'J';
run;
title; footnote;
ods excel close; ods results on;

```


```

/* absolute_column_width = 'number-list'
specifies the column widths to use instead of
allowing SAS to determine them (with measured widths).
Any value of 0 resets that width to the default.
Valid units of measure:
in    inches
cm    centimeters
mm    millimeters
em    standard typesetting unit for width
ex    standard typesetting unit for height
pt    a printer's point
px    pixels
Default    Characters
The unit of Characters works satisfactorily
only when using a fixed-width font
such as Courier New or SAS Monospace. */


```

Demo17 Controlling Column Widths When Using FixedWidth Font
Three Cases, Each in a Separate WorkSheet (ottoms of screen prints are trimmed here)

Case 1: Truncation of content in Column A




Case 2: Width of 7 Character Units Does NOT Accommodate 7-Character Value and 6 Character Units Cramp 6-Character Column Headings



CloseUp of Case 2 Column Headings



Case 3: Best Fit



`%let N = 17; /* Demo Step */`

```

%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 250;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo17 Controlling Column Widths When Using FixedWidth
Font.xlsx"
    options( embedded_titles='yes' embedded_footnotes='yes'
              title_footnote_nobreak='yes' zoom="&ZoomPct"
              sheet_interval='proc' );
title1 justify=left bold
    "Demo&N - Students Whose Name Starts with 'J' - &ZoomPct.% Zoom";
ods excel options( sheet_name='Widths 6,4,4,6,6' zoom="&ZoomPct"
                  absolute_column_width='6,4,4,6,6' );
title2 justify=left bold color=blue
    "Using absolute_column_width='6,4,4,6,6'"
    ' with default width unit of "Characters"';
title3 justify=left bold color=red
    "When using fixed-width font Courier New Bold, the width setting for
Name causes truncation";
title4 justify=left bold color=blue
    "'Characters" as a column width unit IS an accurate gauge for fixed-
width fonts';
proc print data=sashelp.class noobs
    style(data) = {font_face='Courier New' font_weight=bold}
    style(header) = {font_face='Courier New'};
where Name =: 'J';
run;
ods excel options( sheet_name='Widths 7,3,3,6,6' zoom="&ZoomPct"
                  absolute_column_width='7,3,3,6,6' );
title2 justify=left bold color=blue
    "No Truncation with absolute_column_width='7,3,3,6,6'"
    ' for unit of Characters and fixed-width font';
title3 justify=left bold color=red
    "To alleviate this tight fit, instead use
absolute_column_width='8,4,4,7,7'";
proc print data=sashelp.class noobs
    style(data) = {font_face='Courier New' font_weight=bold}
    style(header) = {font_face='Courier New'};
where Name =: 'J';
run;
ods excel options( sheet_name='Widths 8,4,4,7,7' zoom="&ZoomPct"
                  absolute_column_width='8,4,4,7,7' );
title2 justify=left bold color=blue
    "Best Fit when using absolute_column_width='8,4,4,7,7'"
    ' for unit of Characters and fixed-width font';
title3 justify=left bold color=blue

```


```

    "Font is Courier New Bold";
proc print data=sashelp.class noobs
    style(data)    = {font_face='Courier New' font_weight=bold}
    style(header) = {font_face='Courier New'};
where Name =: 'J';
run;
title; footnote;
ods excel close; ods results on;

/* absolute_column_width = 'number-list'
specifies the column widths to use
instead of allowing SAS to determine them (with measured widths).
Any value of 0 resets that width to the default.
Valid units of measure:
in    inches
cm    centimeters
mm    millimeters
em    standard typesetting unit for width
ex    standard typesetting unit for height
pt    a printer's point
px    pixels
Default    Characters
The unit of Characters works satisfactorily
only when using a fixed-width font
such as Courier New or SAS Monospace. */


```

**Demo18 Effect of NOGTITLE NOGFOOTNOTE Options (See, also, Appendix C.)
After clicking on the image to cause the temporary display of its boundary**



```
%let N = 18; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 128;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo18 Effect of NOGTITLE NOGFOOTNOTE Options.xlsx"
  nogtitle nogfootnote /* so that title and footnote text
                        are NOT IN the image */
  options(embedded_titles='yes' embedded_footnotes='yes'
          title_footnote_nobreak='yes'
          sheet_name='Using NOGTITLE NOGFOOTNOTE' zoom="&ZoomPct");
goptions xpixels=1000 ypixels=750;
title1 justify=left bold "Dome&N - "
      color=red "Default Plot with NOGTITLE NOGFOOTNOTE options "
      color=black "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue
      "Text font, size, & color are controlled by ODS Excel and the ODS
      style. All text is outside of the image file.";
title3 justify=left bold color=blue
      "NOGTITLE NOGFOOTNOTE options are on the ODS Excel statement itself,
      not inside its options( . . . ).";
proc gplot data=sashelp.class; plot weight*height; run; quit;
title; footnote;
ods excel close; ods results on;
```

Demo19 Effect of GTITLE GFOOTNOTE Options
After clicking on the image to cause the temporary display of its boundary



```

%let N = 19; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 160;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods excel file="&Path\Demo19 Effect of GTITLE GFOOTNOTE Options.xlsx"
  gtitle gfootnote /* So that title and footnote text
                    are IN the image. These are the defaults. */
  options(embedded_titles='yes' embedded_footnotes='yes'
          title_footnote_nobreak='yes'
          sheet_name='Using GTITLE GFOOTNOTE' zoom="&ZoomPct");
goptions xpixels=1000 ypixels=750;
title1 justify=left bold "Dome&N - "
      color=red "Default Plot with GTITLE GFOOTNOTE options "
      color=black "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue
      "Text font, size, and color controlled by graph code. All text is
inside the image file.";
title3 justify=left bold color=blue
      "GTITLE GFOOTNOTE options are on the ODS Excel statement, not inside
its options( . . . ).";
proc gplot data=sashelp.class; plot weight*height; run; quit;
title; footnote; ods excel close; ods results on;

```


Demo20 PROC UNIVARIATE Outputs in Separate WorkSheets
Each of five worksheets & display of five sheetnames cropped and enlarged.

Variable: Weight

Moments			
N	19	Sum Weights	19
Mean	100.026316	Sum Observations	1900.5
Std Deviation	22.7739335	Variance	518.652047
Skewness	0.18335097	Kurtosis	0.68336484
Uncorrected SS	198435.75	Corrected SS	9335.73684
Coeff Variation	22.7679419	Std Error Mean	5.22469887

Variable: Weight

Basic Statistical Measures			
Location		Variability	
Mean	100.0263	Std Deviation	22.77393
Median	89.5000	Variance	518.65205
Mode	84.0000	Range	99.50000
		Interquartile Range	28.50000

Note: The mode displayed is the smallest of 4 modes with a count of 2.

Variable: Weight


Tests for Location: Mu=0			
Test	Statistic	p Value	
Student's t	t	19.1440	Pr > t < .0001
Sign	M	9.5	Pr >= M < .0001
Signed Rank	S	95	Pr >= S < .0001

Variable: Weight

Quantiles (Definition 5)	
Level	Quantile
100% Max	150.0
99%	150.0
95%	150.0
90%	133.0
75% Q3	112.5
50% Median	99.5
25% Q1	84.0
10%	77.0
5%	50.5
1%	50.5
0% Min	50.5

Variable: Weight

Extreme Observations			
Lowest		Highest	
Value	Obs	Value	Obs
50.5	11	112.5	1
77.0	13	112.5	8
83.0	6	128.0	16
84.0	9	133.0	17
84.0	2	150.0	15



/* SheetNames after the first get a sequence number appended */

```

%let N = 20; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
    
```



```

%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 200;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods noproctitle; /* suppress identification of PROC in the output */
ods excel file="&Path\Demo20 PROC UNIVARIATE Outputs in Separate
WorkSheets.xlsx"
    options( embedded_titles='yes' embedded_footnotes='yes'
            title_footnote_nobreak='yes' sheet_name='Univariate_Output'
            sheet_interval='output' zoom="&ZoomPct" );
title1 justify=left bold
    "Demo&N - "
    color=red
    "Univariate Statistics for Student Weight "
    color=black
    "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue
    "In Five Separate WorkSheets - using option sheet_interval='output'";
options nocenter; /* Variable: varname at left,
                    rather than oddly far out to the right.
    Tabular output & text lines are by default left-justified. */
proc univariate data=sashelp.class;
var weight;
run;
options center; /* restore default */
title; footnote;
ods excel close; ods results on;

```

Demo21 PROC UNIVARIATE Outputs in the Same WorkSheet

The screenshot shows the following tables in the spreadsheet:

- Variable Weight:** A table with columns for Variable, N, Mean, Std Deviation, and Std Error Mean. It lists variables like WGT, WGT2, WGT3, WGT4, WGT5, WGT6, WGT7, WGT8, WGT9, WGT10, WGT11, WGT12, WGT13, WGT14, WGT15, WGT16, WGT17, WGT18, WGT19, WGT20, WGT21.
- Basic Statistical Measures:** A table with columns for Variable, Location, Scale, and Skewness. It lists variables like WGT, WGT2, WGT3, WGT4, WGT5, WGT6, WGT7, WGT8, WGT9, WGT10, WGT11, WGT12, WGT13, WGT14, WGT15, WGT16, WGT17, WGT18, WGT19, WGT20, WGT21.
- Tests for Location Shift:** A table with columns for Test, Statistic, and P-Value. It lists tests like T-TEST, SIGNED RANK, and SIGN RANK.
- Skewed Distribution:** A table with columns for Variable, Statistic, and P-Value. It lists statistics like SKEWNESS, KURTOSIS, and MODALITY.
- Extreme Observations:** A table with columns for Variable, Value, and Weight. It lists extreme values for variables like WGT, WGT2, WGT3, WGT4, WGT5, WGT6, WGT7, WGT8, WGT9, WGT10, WGT11, WGT12, WGT13, WGT14, WGT15, WGT16, WGT17, WGT18, WGT19, WGT20, WGT21.

```

%let N = 21; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 85;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
ods results off; ods _all_ close;
ods noproctitle; /* suppress identification of PROC in the output */
ods excel file="&Path\Demo21 PROC UNIVARIATE Outputs in the Same
WorkSheet.xlsx"
    options( embedded_titles='yes' embedded_footnotes='yes'
            title_footnote_nobreak='yes'
            zoom="&ZoomPct" sheet_interval='none'
            sheet_name='All Outputs On Same Sheet' );
title1 justify=left bold "Demo&N - "
    color=red "Univariate Statistics for Student Weight "
    color=black "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue
    "All On Same WorkSheet - using option sheet_interval='none'";
options nocenter; /* To get Variable: varname at left, rather than oddly far
out to right. Tabular output & text lines are by default left-justified. */
proc univariate data=sashelp.class;
var weight;
run;
options center; /* restore default */
title; footnote;
ods excel close; ods results on;

```

Demo22 Two Tables Stacked

The screenshot shows an Excel spreadsheet with two tables stacked vertically. The first table is titled 'Girls' and contains 8 rows of data. The second table is titled 'Boys' and also contains 8 rows of data. Both tables have columns for Name, Age, Height, and Weight. The spreadsheet is titled 'Demo22 - Two Tables Stacked - 140% Zoom'.

Girls			
Name	Age	Height	Weight
Alice	13	65.5	84.0
Barbara	13	65.3	98.0
Carl	14	62.8	102.5
Jane	12	59.8	84.5
Janet	15	62.5	112.5
Joyce	11	51.3	50.5
Judy	14	64.3	90.0
Lois	12	56.3	77.0
Mary	15	66.5	112.0

Boys			
Name	Age	Height	Weight
Athel	14	69.0	112.5
Henry	14	63.5	102.5
James	12	57.3	83.0
Jeffrey	13	62.5	84.0
John	12	59.0	59.5
Philip	16	72.0	150.0
Robert	12	64.8	128.0
Ronald	15	67.0	133.0
Thomas	11	57.5	65.0
William	15	66.5	112.0

Source Data: SASHELP.CLASS
Code: C:\MWSUG_2019\ODS_Excel\code\Demo22.sas
Run on: Wednesday, 31 July 2019 at 8:56:47 PM

```

%let N = 22; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 140;
ods results off; ods _all_ close;
ods excel file="&Path\Demo22 Two Tables Stacked.xlsx"
  options( embedded_titles='yes' embedded_footnotes='yes'
    zoom="&ZoomPct" sheet_name='Two Tables Stacked'
    sheet_interval='none' title_footnote_nobreak='yes' );
title1 justify=left bold color=black "Demo&N - "
  color=red "Two Tables Stacked "
  color=black "- &ZoomPct.% Zoom ";
title2 ' ';
title3 bold justify=left 'Girls';
footnote;
proc print data=sashelp.class noobs;
where Sex='F';
var Name Age Height Weight;
run;
title1 bold justify=left 'Boys';
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
proc print data=sashelp.class noobs;
where Sex='M';
var Name Age Height Weight;
run;
title; footnote; ods excel close; ods results on;

```

**Demo23 Two Tables Side By Side Using PROC PRINT
(This is really one table.)**

The screenshot shows an Excel spreadsheet with the following content:

1 **Demo23 - Two Tables Side By Side - 210% Zoom**
 2 (This is really one table. Using PROC PRINT.)
 3
 4 **Girls** **Boys**
 5
 6 **Name** **Age** **Height** **Weight** **Name** **Age** **Height** **Weight**
 7 Alice 13 56.5 84.0 Alfred 14 69.0 112.5
 8 Barbara 13 65.3 98.0 Henry 14 63.5 102.5
 9 Carol 14 62.8 102.5 James 12 57.3 83.0
 10 Jane 12 59.8 84.5 Jeffrey 13 62.5 84.0
 11 Janet 15 62.5 112.5 John 12 59.0 99.5
 12 Joyce 11 51.3 50.5 Philip 16 72.0 150.0
 13 Judy 14 64.3 90.0 Robert 12 64.8 128.0
 14 Louise 12 56.3 77.0 Ronald 15 67.0 133.0
 15 Mary 15 66.5 112.0 Thomas 11 57.5 85.0
 16 William 15 66.5 112.0
 17
 18 **Source Data: SASHELP.CLASS**
 19 **Code: C:\MWSUG_2019\ODS_Excel\code\Demo23.sas**
 20 **Run on: Wednesday, 31 July 2019 at 8:59:49 PM**
 21

```
/* This is really one table. Using PROC PRINT. */
/* Except for using a control for the TITLE width, there is nothing special
here about use of ODS Excel. It is really a Base SAS programming trick to
overcome the fact that ODS Excel does not permit the placement of output from
two PROC runs side by side in a worksheet. */
```

```
%let N = 23; /* Demo Step */
%let Path = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 210;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
data
work.Girls
(keep=MergeSeqNo
Name_F Age_F Height_F Weight_F)
work.Boys
(keep=MergeSeqNo
Name_M Age_M Height_M Weight_M);
retain CountGirls CountBoys 0;
set sashelp.class;
if Sex EQ 'F'
then do;
CountGirls + 1;
Name_F=Name;
Age_F=Age;
Height_F=Height;
```

```

    Weight_F=Weight;
    MergeSeqNo = CountGirls;
    output work.Girls;
end;
else do;
    CountBoys + 1;
    Name_M=Name;
    Age_M=Age;
    Height_M=Height;
    Weight_M=Weight;
    MergeSeqNo = CountBoys;
    output work.Boys;
end;
run;
data work.ToPrint;
retain TableSeparator ' ';
merge work.Girls(in=F) work.Boys(in=M);
by MergeSeqNo;
if F or M;
run;
ods results off; ods _all_ close;
ods excel file="&Path\Demo23 Two Tables Side By Side Using PROC PRINT.xlsx"
    options( embedded_titles='yes' embedded_footnotes='yes'
        zoom="&ZoomPct" sheet_name='Two Tables On Same WorkSheet'
        title_footnote_width='9' );
title1 justify=left bold color=black "Demo&N - "
    color=red
    "Two Tables Side By Side "
    color=black "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue "(This is really one table. Using PROC
PRINT.)";
title3 ' ';
title4 bold justify=center
'Girls                                     Boys';
/* NOTE: The precise space inside TITLE3 text is essential for alignment of
the table headings. */
options missing=' ';
proc print data=work.ToPrint noobs label;
var
Name_F Age_F Height_F Weight_F
TableSeparator
Name_M Age_M Height_M Weight_M;
label Name_F='Name';
label Age_F='Age';
label Height_F='Height';
label Weight_F='Weight';
label TableSeparator='00'X;
label Name_M='Name';
label Age_M='Age';
label Height_M='Height';
label Weight_M='Weight';
run;
title; footnote;
ods excel close; ods results on;

```

**Demo24 Two Tables Side By Side Using PROC REPORT
(This is really one table.)**

Demo24 - Two Tables Side By Side - 230% Zoom

(This is really one table. Using PROC REPORT.)

Sales By Region		Sales By Product	
Region	Sales	Product	Sales
Africa	\$2,342,588	Boot	\$2,350,543
Asia	\$460,231	Men's Casual	\$7,933,707
Canada	\$4,255,712	Men's Dress	\$5,507,243
Central America/Caribbean	\$3,657,753	Sandal	\$868,436
Eastern Europe	\$2,394,940	Slipper	\$6,175,834
Middle East	\$5,631,779	Sport Shoe	\$651,467
Pacific	\$2,296,794	Women's Casual	\$4,137,861
South America	\$2,434,783	Women's Dress	\$6,226,475
United States	\$5,503,986		
Western Europe	\$4,873,000		

Source Data: SASHELP.SHOES
Code: C:\MWSUG_2019\ODS_Excel\code\Demo24.sas
Run on: Wednesday, 31 July 2019 at 8:59:51 PM

/* This is really one table. Using PROC REPORT. */
/* There is nothing special here about use of ODS Excel. It is really a Base SAS programming trick to overcome the fact that ODS Excel does not permit the placement of output from two PROC runs side by side in a worksheet. */

```
%let N = 24; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 230;
footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
```

```
proc summary data=sashelp.shoes nway;
class Region; var sales;
output out=work.SalesByRegion sum=RegionSales;
run;
```

```
proc summary data=sashelp.shoes nway;
class Product; var sales;
output out=work.SalesByProduct sum=ProductSales;
run;
```

```
data Table1;
  set work.SalesByRegion;
  ordno + 1;
run;
```

```
data Table2;
```



```

    set work.SalesByProduct;
    ordno + 1;
run;

data final;
    merge Table1(in=in1) Table2(in=in2);
    by ordno;
    if in1 or in2;
run;

ods results off; ods _all_ close;
ods excel file="&Path\Demo24 Two Tables Side By Side Using PROC REPORT.xlsx"
    options( embedded_titles='yes' embedded_footnotes='yes'
            zoom="&ZoomPct" sheet_name='Tables Side By Side'
            title_footnote_nobreak='yes' );
title1 justify=left bold color=black "Demo&N - "
    color=red
    "Two Tables Side By Side "
    color=black "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue "(This is really one table. Using PROC
REPORT.)";
options missing=' ';
proc report data=final;
    column ordno ('Sales By Region' Region RegionSales )
           blank
           ('Sales By Product' Product ProductSales);
    define RegionSales / display 'Sales';
    define ProductSales / display 'Sales';
    define ordno / order noprint;
    define blank / computed ' ' style=Header{width=.1in};
    compute blank /character length=1;
        blank = ' ';
    endcomp;
run;
title; footnote;
ods excel close; ods results on;

```

Demo25 Four Tables Side By Side Using Excel_Enhance

The screenshot shows an Excel spreadsheet with four tables side-by-side. The tables are:

- Ranked Sales By Region:**

Region	Sales
Middle East	\$5,631,779
United States	\$5,603,986
Western Europe	\$4,873,000
Canada	\$4,255,712
Central America/Caribbean	\$3,657,753
South America	\$2,434,783
Eastern Europe	\$2,394,940
Africa	\$2,342,588
Pacific	\$2,296,794
Asia	\$460,231
- Ranked Sales By Product:**

Product	Sales
Men's Casual	\$7,933,707
Women's Dress	\$6,226,475
Slipper	\$6,178,834
Men's Dress	\$5,507,243
Women's Casual	\$4,137,861
Boot	\$2,350,543
Sandal	\$868,436
Sport Shoe	\$651,467
- Top Ten City Sales Ranked Across All Regions:**

City	Region	Sales
Vancouver	Canada	\$3,227,768
Tel Aviv	Middle East	\$2,567,568
Kingston	Central America/Caribbean	\$2,235,204
Dubai	Middle East	\$1,910,544
Chicago	United States	\$1,565,585
New York	United States	\$1,489,207
Al-Khobar	Middle East	\$1,153,667
Minneapolis	United States	\$1,099,937
Heidelberg	Western Europe	\$967,739
Lisbon	Western Europe	\$898,345
- Top City Sales In Each Region:**

City	Region	Sales
Vancouver	Canada	\$3,227,768
Tel Aviv	Middle East	\$2,567,568
Kingston	Central America/Caribbean	\$2,235,204
Chicago	United States	\$1,565,585
Heidelberg	Western Europe	\$967,739
Manila	Pacific	\$854,904
Caracas	South America	\$789,323
Warsaw	Eastern Europe	\$786,714
Cairo	Africa	\$738,188
Seoul	Asia	\$442,409

Source Data: SASHELP.SHOES
Code: C:\MWSUG_2019\ODS_Excel\codeDemo25.sas
Run on: Wednesday, 31 July 2019 at 6:30:46 PM

```

%let N = 25; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\Excel_Enhance.sas";
%include "&CodePath.\macros\CreateTableAsImageFile.sas";
%include "&CodePath.\macros\BlankRowInWorkSheet.sas";
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
options noxwait noxsync;
options mprint;
%let ZoomPct = 130;

title; footnote;
ods results off; ods _all_ close;

ods excel file="c:\temp\Temp_Demo&N..xlsx"
           options( zoom="&ZoomPct"
                   sheet_name='FourTablesSideBySide' );

proc odstext;
p "Demo&N - Four Tables Side By Side - &ZoomPct.% Zoom"
  / style={just=left font=(Arial) fontsize=12pt fontweight=Bold};
%BlankRowInWorkSheet(Count=18,Font=Arial,FontSize=12pt,FontWeight=Bold);
p 'Source Data: SASHELP.SHOES'
  / style={just=left font=(Arial) fontsize=12pt fontweight=Bold};
p "Code: &CodePath.Demo&N..sas"
  / style={just=left font=(Arial) fontsize=12pt fontweight=Bold};

```

```
p "Run on: &RunDayDateTime"  
  / style={just=left font=(Arial) fontsize=12pt fontweight=Bold};  
run;
```

```
title; footnote;  
ods excel close;
```

```
proc summary data=sashelp.shoes nway;  
class region;  
var sales;  
output out=SalesByRegion(drop=_type__freq_) sum=Sales;  
run;
```

```
proc sort data=SalesByRegion;  
by descending Sales;  
run;
```

```
proc summary data=sashelp.shoes nway;  
class product;  
var sales;  
output out=SalesByProduct(drop=_type__freq_) sum=Sales;  
run;
```

```
proc sort data=SalesByProduct;  
by descending Sales;  
run;
```

```
proc summary data=sashelp.shoes nway;  
class subsidiary;  
id region;  
var sales;  
output out=SalesByCity(drop=_type__freq_) sum=Sales;  
run;
```

```
proc sort data=SalesByCity out=RankedByCitySales;  
by descending Sales;  
run;
```

```
data TopTenCitySalesInAllRegions(drop=Subsidiary);  
length City $ 12;  
set RankedByCitySales;  
City = Subsidiary;  
output;  
if _N_ EQ 10 then stop;  
run;
```

```
proc sort data=SalesByCity out=RankedByCitySalesInRegion;  
by Region descending Sales;  
run;
```

```
data TopCitySalesInEachRegion(drop=Subsidiary);  
length City $ 12;  
set RankedByCitySalesInRegion;  
by Region;  
if first.Region;
```

```

City = Subsidiary;
run;

proc sort data=TopCitySalesInEachRegion;
by descending Sales;
run;

options nocenter nonumber nodate;

%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,
    data=SalesByRegion,
    title=Ranked Sales By Region,
    width_inches=4,
    height_inches=4);

%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,
    data=SalesByProduct,
    title=Ranked Sales By Product,
    width_inches=4,
    height_inches=4);


%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,
    data=TopTenCitySalesInAllRegions,
    title=Top Ten City Sales Ranked Across All Regions,
    width_inches=5,
    height_inches=4);

%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,
    data=TopCitySalesInEachRegion,
    title=Top City Sales In Each Region,
    width_inches=5,
    height_inches=4);

%Excel_Enhance(
    open_workbook=c:\temp\Temp_Demo&N..xlsx,
    insert_image=%str(
c:\temp\Demo&N._SalesByRegion.png#FourTablesSideBySide!a3,
c:\temp\Demo&N._SalesByProduct.png#FourTablesSideBySide!e3,
c:\temp\Demo&N._TopTenCitySalesInAllRegions.png#FourTablesSideBySide!h3,
c:\temp\Demo&N._TopCitySalesInEachRegion.png#FourTablesSideBySide!n3),
    create_workbook=
        &Path\Demo25 Four Tables Side By Side Using Excel_Enhance.xlsx,
    file_format=xlsx);

```

Demo26 Four Tables in a Two By Two Array Using Excel Enhance



```

%let N = 26; /* Demo Step */
/* BELOW, the code parts that are highlighted
are those that differ from Demo25 */

%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\Excel_Enhance.sas";
%include "&CodePath.\macros\CreateTableAsImageFile.sas";
%include "&CodePath.\macros\BlankRowInWorkSheet.sas";
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
options noxwait noxsync;
options mprint;
%let ZoomPct = 97;

title; footnote;
ods results off;
ods _all_ close;

ods excel file="c:\temp\Temp_Demo&N..xlsx"
options( zoom="&ZoomPct"
sheet_name='FourTablesInTwoByTwoArray' );

proc odstext;
p "Demo&N - Four Tables in a Two By Two Array - &ZoomPct.% Zoom"
/ style={just=left font=(Arial) fontsize=12pt fontweight=Bold};
%BlankRowInWorkSheet (Count=43,Font=Arial,FontSize=9.5pt,FontWeight=Bold);
p 'Source Data: SASHELP.SHOES'
/ style={just=left font=(Arial) fontsize=12pt fontweight=Bold};
p "Code: &CodePath.Demo&N..sas"
/ style={just=left font=(Arial) fontsize=12pt fontweight=Bold};
p "Run on: &RunDayDateTime"

```

```

    / style={just=left font=(Arial) fontsize=12pt fontweight=Bold};
run;

title; footnote;
ods excel close;

proc summary data=sashelp.shoes nway;
class region;
var sales;
output out=SalesByRegion(drop=_type_ _freq_) sum=Sales;
run;

proc sort data=SalesByRegion;
by descending Sales;
run;

proc summary data=sashelp.shoes nway;
class product;
var sales;
output out=SalesByProduct(drop=_type_ _freq_) sum=Sales;
run;

proc sort data=SalesByProduct;
by descending Sales;
run;

proc summary data=sashelp.shoes nway;
class subsidiary;
id region;
var sales;
output out=SalesByCity(drop=_type_ _freq_) sum=Sales;
run;

proc sort data=SalesByCity out=RankedByCitySales;
by descending Sales;
run;

data TopTenCitySalesInAllRegions(drop=Subsidiary);
length City $ 12;
set RankedByCitySales;
City = Subsidiary;
output;
if _N_ EQ 10 then stop;
run;

proc sort data=SalesByCity out=RankedByCitySalesInRegion;
by Region descending Sales;
run;

data TopCitySalesInEachRegion(drop=Subsidiary);
length City $ 12;
set RankedByCitySalesInRegion;
by Region;
if first.Region;
City = Subsidiary;
run;

```



```

proc sort data=TopCitySalesInEachRegion;
by descending Sales;
run;

options nocenter nonumber nodate;

%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,
    data=SalesByRegion,title=Ranked Sales By Region,
    width_inches=3,height_inches=3.75);

%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,
    data=SalesByProduct,
    title=Ranked Sales By Product,
    width_inches=3,height_inches=3.75);


%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,
    data=TopTenCitySalesInAllRegions,
    title=Top Ten City Sales Ranked Across All Regions,
    width_inches=5,height_inches=3.75);

%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,
    data=TopCitySalesInEachRegion,
    title=Top City Sales In Each Region,
    width_inches=5,height_inches=3.75);

%Excel_Enhance(
    open_workbook=c:\temp\Temp_Demo&N..xlsx,
    insert_image=%str(
c:\temp\Demo&N._SalesByRegion.png#FourTablesInTwoByTwoArray!a3,
c:\temp\Demo&N._TopTenCitySalesInAllRegions.png#FourTablesInTwoByTwoArray!e3,
c:\temp\Demo&N._SalesByProduct.png#FourTablesInTwoByTwoArray!a24,
c:\temp\Demo&N._TopCitySalesInEachRegion.png#FourTablesInTwoByTwoArray!e24
    ),
    create_workbook=
&Path\Demo26 Four Tables in a Two By Two Array Using Excel_Enhance.xlsx,
    file_format=xlsx);

```

Demo27 Two Plots Stacked



```

%let N = 27; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 80;

ods results off; ods _all_ close;

ods excel file="&Path\Demo27 Two Plots Stacked.xlsx"
  nogtitle /* titles outside of graph image files */
  nogfootnote /* footnotes outside of graph image files */
  options( sheet_name='Two Plots Stacked'
    sheet_interval='none'
    embedded_titles='yes'
    embedded_footnotes='yes'
    title_footnote_nobreak='yes'
    zoom="&ZoomPct" );

ods graphics on / reset=all scale=off width=4in height=4in;
title1 justify=left font=Arial bold height=13pt /* See Appendix C */
  "Demo&N - "
  color=red height=13pt /* See Appendix C. */
  "Two Plots Stacked - "
  color=black height=13pt /* See Appendix C */
  "&ZoomPct.% Zoom";
/* If you wish, remove height=13pt for the red and blue parts
to see what happens. */

```

```


title2 justify=left font=Arial height=8pt bold ' ';
title3 justify=left font=Arial height=13pt bold
  'Student Height (inches) vs Age';
footnote;
proc sgplot data=sashelp.class noborder;
scatter y=height x=age /
  datalabel
  datalabelattrs=(family=Arial size=13pt weight=bold)
  FilledOutlinedMarkers
  markerattrs=(symbol=CircleFilled size=6pt)
  markerfillattrs=(color=red)
  markeroutlineattrs=(color=blue thickness=1);
yaxis display=none;
xaxis display=(nolabel noline noticks)
  valueattrs=(family=Arial size=13pt weight=bold);
run;

title1 justify=left font=Arial height=13pt bold
  'Student Weight (pounds) vs Age';
footnote1 justify=left font=Arial height=13pt bold "Source Data:
SASHELP.CLASS";
footnote2 justify=left font=Arial height=13pt bold "Code:
&CodePath\Demo&N..sas";
footnote3 justify=left font=Arial height=13pt bold bold "Run on:
&RunDayDate'Time";
proc sgplot data=sashelp.class noborder;
scatter y=weight x=age /
  datalabel
  datalabelattrs=(family=Arial size=13pt weight=bold)
  FilledOutlinedMarkers
  markerattrs=(symbol=CircleFilled size=6pt)
  markerfillattrs=(color=red)
  markeroutlineattrs=(color=blue thickness=1);
yaxis display=none;
xaxis display=(nolabel noline noticks)
  valueattrs=(family=Arial size=13pt weight=bold);
run;

title; footnote;
ods excel close; ods results on;

```

Demo28 Two Plots Side By Side Using Excel_Enhance



```

%let N = 28; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\Excel_Enhance.sas";
%include "&CodePath.\macros\CreateTableAsImageFile.sas";
%include "&CodePath.\macros\BlankRowInWorkSheet.sas";
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
options noxwait noxsync;
options mprint;
%let ZoomPct = 160;

title; footnote;
ods results off; ods _all_ close;

ods excel file="c:\temp\Temp_Demo&N..xlsx"
           options( zoom="&ZoomPct"
                   sheet_name='TwoPlotsSideBySide' );

proc odstext;
p "Demo&N - Two Plots Side By Side - &ZoomPct.% Zoom"
  / style={just=left font=(Arial) fontsize=11pt fontweight=Bold};
%BlankRowInWorkSheet(Count=23,Font=Arial,FontSize=9.5pt,FontWeight=Bold);
p 'Source Data: SASHELP.CLASS'
  / style={just=left font=(Arial) fontsize=9.5pt fontweight=Bold};
p "Code: &CodePath.Demo&N..sas"
  / style={just=left font=(Arial) fontsize=9.5pt fontweight=Bold};
p "Run on: &RunDayDateTime"
  / style={just=left font=(Arial) fontsize=9.5pt fontweight=Bold};
run;

```

```

title; footnote;
ods excel close;

ods listing gpath="c:\temp";

ods graphics on / reset=all scale=off width=4in height=4in
imagename="Demo&N._Left";
title font='Arial/Bold' height=14pt
  'Student Height (inches) vs Age';
footnote;
proc sgplot data=sashelp.class noborder;
scatter y=height x=age /
  datalabel
  datalabelattrs=(family=Arial size=12pt weight=bold)
  FilledOutlinedMarkers
  markerattrs=(symbol=CircleFilled size=9pt)
  markerfillattrs=(color=red)
  markeroutlineattrs=(color=blue thickness=1);
yaxis display=none;
xaxis display=(nolabel noline noticks)
  valueattrs=(family=Arial size=12pt weight=bold);
run;


ods graphics on / reset=all scale=off width=4in height=4in
imagename="Demo&N._Right";
title font='Arial/Bold' height=14pt
  'Student Weight (pounds) vs Age';
footnote;
proc sgplot data=sashelp.class noborder;
scatter y=weight x=age /
  datalabel
  datalabelattrs=(family=Arial size=12pt weight=bold)
  FilledOutlinedMarkers
  markerattrs=(symbol=CircleFilled size=9pt)
  markerfillattrs=(color=red)
  markeroutlineattrs=(color=blue thickness=1);
yaxis display=none;
xaxis display=(nolabel noline noticks)
  valueattrs=(family=Arial size=12pt weight=bold);
run;

title; footnote; ods listing close;

%Excel_Enhance(
  open_workbook=c:\temp\Temp_Demo&N..xlsx,
  insert_image=%str(c:\temp\Demo&N._Left.png#TwoPlotsSideBySide!a3,
    c:\temp\Demo&N._Right.png#TwoPlotsSideBySide!f3),
  create_workbook=
    &Path\Demo28 Two Plots Side By Side Using Excel_Enhance.xlsx,
  file_format=xlsx);

```

Demo29 Two Plots & Two Tables Using Excel Enhance



```

%let N = 29; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\Excel_Enhance.sas";
%include "&CodePath.\macros\CreateTableAsImageFile.sas";
%include "&CodePath.\macros\BlankRowInWorkSheet.sas";
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
options noxwait noxsync;
options mprint;
%let ZoomPct = 90;

title; footnote;
ods results off; ods _all_ close;

ods excel file="c:\temp\Temp_Demo&N..xlsx"
           options(zoom="&ZoomPct"
                  sheet_name='TwoPlotsTwoTablesIn2X2Array');

proc odstext;
p "Demo&N - Two Plots and Two Tables in a Two By Two Array - &ZoomPct.% Zoom"
  / style={just=left font=(Arial) fontsize=13pt fontweight=Bold};
%BlankRowInWorkSheet(Count=46,Font=Arial,FontSize=9.5pt,FontWeight=Bold);
p 'Source Data: SASHELP.CLASS'
  / style={just=left font=(Arial) fontsize=10pt fontweight=Bold};
p "Code: &CodePath.Demo&N..sas"
  / style={just=left font=(Arial) fontsize=10pt fontweight=Bold};
p "Run on: &RunDayDateTime"
  / style={just=left font=(Arial) fontsize=10pt fontweight=Bold};
run;

```



```

title; footnote;
ods excel close;

data StudentNameHgtAge;
set sashelp.class;
keep Name Age Height;
run;

data StudentNameWgtAge;
set sashelp.class;
keep Name Age Weight;
run;

ods listing gpath="c:\temp";

ods graphics on / reset=all scale=off width=4in height=4in
imagenam="Demo&N._HeightVsAge";
title font='Arial/Bold' height=12pt
'Student Height (inches) vs Age';
footnote;
proc sgplot data=StudentNameHgtAge noborder;
scatter y=height x=age /
    datalabel
    datalabelattrs=(family=Arial size=12pt)
    FilledOutlinedMarkers
    markerattrs=(symbol=CircleFilled size=9pt)
    markerfillattrs=(color=red)
    markeroutlineattrs=(color=blue thickness=1);
yaxis display=none;
xaxis display=(nolabel noline noticks)
    valueattrs=(family=Arial size=12pt weight=bold);
run;

ods graphics on / reset=all scale=off width=4in height=4in
imagenam="Demo&N._WeightVsAge";
title font='Arial/Bold' height=12pt
'Student Weight (pounds) vs Age';
footnote;
proc sgplot data=StudentNameWgtAge noborder;
scatter y=weight x=age /
    datalabel
    datalabelattrs=(family=Arial size=12pt)
    FilledOutlinedMarkers
    markerattrs=(symbol=CircleFilled size=9pt)
    markerfillattrs=(color=red)
    markeroutlineattrs=(color=blue thickness=1);
yaxis display=none;
xaxis display=(nolabel noline noticks)
    valueattrs=(family=Arial size=12pt weight=bold);
run;
title; footnote;
ods listing close;

options nocenter nonumber nodate;

%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,

```

```


    data=StudentNameHgtAge,
    title=Student Ages And Heights,
    width_inches=2.2,height_inches=4);

%CreateTableAsImageFile(
    imagefilenameprefix=Demo&N._,
    data=StudentNameWgtAge,
    title=Student Ages And Weights,
    width_inches=2.2,height_inches=4);

%Excel_Enhance(
    open_workbook=c:\temp\Temp_Demo&N..xlsx,
    insert_image=%str(
    c:\temp\Demo&N._HeightVsAge.png#TwoPlotsTwoTablesIn2X2Array!a3,
    c:\temp\Demo&N._StudentNameHgtAge.png#TwoPlotsTwoTablesIn2X2Array!f3,
    c:\temp\Demo&N._WeightVsAge.png#TwoPlotsTwoTablesIn2X2Array!a26,
    c:\temp\Demo&N._StudentNameWgtAge.png#TwoPlotsTwoTablesIn2X2Array!f26
    ),
    create_workbook=
        &Path\Demo29 Two Plots & Two Tables Using Excel_Enhance.xlsx,
    file_format=xlsx);

```

Demo30 Bar Chart & Table in Separate Worksheets



All Information about the Students

Name	Sex	Age	Height	Weight
Alfred	M	14	69.0	112.5
Alice	F	13	56.5	84.0
Barbara	F	13	65.3	98.0
Carol	F	14	62.8	102.5
Henry	M	14	63.5	102.5
James	M	12	57.3	83.0
Jane	F	12	59.8	84.5
Janet	F	15	62.5	112.5
Jeffrey	M	13	62.5	84.0
John	M	12	59.0	98.5
Joyce	F	11	51.3	90.5
Judy	F	14	64.3	96.0
Louise	F	12	56.3	77.0
Mary	F	15	66.5	112.0
Philip	M	16	72.0	156.0
Robert	M	12	64.8	128.0
Ronald	M	15	67.0	135.0
Thomas	M	11	57.5	85.0
William	M	15	66.5	112.0

Source Data: SASHELP.CLASS
Code: C:\MWSUG_2019\ODS_Excelcode\Demo30.sas
Run on: Wednesday, 31 July 2019 at 8:59:57 PM

```

%let N = 30; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
footnote1 justify=left bold height=14pt "Source Data: SASHELP.CLASS";
footnote2 justify=left bold height=14pt "Code: &CodePath\Demo&N..sas";

```

```

footnote3 justify=left bold height=14pt "Run on: &RunDayDateTime";

ods results=off; ods _all_ close;
ods excel file= "&Path\Demo30 Bar Chart & Table in Separate WorkSheets.xlsx"
  nogtitle      /* display the graph titles above the graph image
                 on that worksheet */
  nogfootnote  /* display the graph footnotes below the graph image
                 on that worksheet */
  options( embedded_titles='yes' embedded_footnotes='yes'
           title_footnote_nobreak='yes'
           zoom="145"
           sheet_interval='proc' );


ods excel options( sheet_name='Ranked Height of Students');

ods graphics / reset=all scale=off width=8.0in height=4.0in;
title1 height=14pt font='Arial/Bold' justify=left "Demo&N - "
  color=red height=14pt /* See Appendix C */
  "Bar Chart and Table in Separate Worksheets "
  color=black height=14pt /* See Appendix C */
  "- 145% Zoom";
title2 height=14pt font='Arial/Bold' justify=left
  "Ranked Height of Students";
proc sgplot data=sashelp.class noborder;
vbar name /
  response=height
  categoryorder=respdesc
  datalabel
  datalabelattrs=(family='Arial' size=12pt weight=Bold)
  fillattrs=(color=CX009900)
  nooutline
  displaybaseline=off
  barwidth=0.4;
yaxis display=none;
xaxis display=(noticks noline nolabel)
  valueattrs=(family='Arial' size=12pt weight=Bold)
  fitpolicy=stagger;
run;

ods excel options(zoom="120" sheet_name='All Information for Students');
title1 height=14pt font='Arial/Bold' justify=left "Demo&N - "
  color=red height=14pt /* See Appendix C */
  "Bar Chart and Table in Separate Worksheets "
  color=black height=14pt /* See Appendix C */
  "- 120% Zoom";
title2 font='Arial/Bold' height=14pt justify=left
  "All Information about the Students";
proc print data=sashelp.class noobs
  /* force Arial instead of Calibri, and control fontsize */
  style(header)=[font=(Arial) fontsize=12pt fontweight=Bold]
  style(data) =[font=(Arial) fontsize=12pt fontweight=Bold];
run;
title; footnote;
ods excel close; ods results=on;

```

Demo31 Bar Chart Over Table in the Same WorkSheet



```
%let N = 31; /* Demo Step */%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath  = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 115;
```

```
ods results=off; ods _all_ close;
```

```
ods excel file= "&Path\Demo31 Bar Chart Over Table in the Same
WorkSheet.xlsx"
```

```
options( embedded_titles='yes' embedded_footnotes='yes'
        title_footnote_nobreak='yes'
        zoom="&ZoomPct"
        sheet_interval='none'
        sheet_name='Bar Chart Over Table' );
```

```
ods graphics / reset=all scale=off width=6.0in height=3.5in;
title1 height=14pt font='Arial/Bold' justify=left "Demo&N - "
        color=red height=14pt
        "Bar Chart and Table Stacked "
        color=black height=14pt
        "- &ZoomPct.% Zoom";
```

```
title2 ' ';
title3 height=14pt font='Arial/Bold' justify=left "Ranked Height of Students
Whose Names Start with 'J'";
```

```
footnote;
```

```
proc sgplot data=sashelp.class noborder;
```

```
where name =: 'J';
```

```
vbar name /
```

```
response=height
```

```
categoryorder=respdesc
```

```


datalabel
datalabelattrs=(family='Arial' size=12pt weight=Bold)
fillattrs=(color=CX009900)
nooutline
displaybaseline=off
barwidth=0.4;
yaxis display=none;
xaxis display=(noticks noline nolabel)
valueattrs=(family='Arial' size=12pt weight=Bold)
fitpolicy=stagger;
run;

title1 font='Arial/Bold' height=15pt justify=left
  "All Information about the Students Whose Names Start with 'J'";
footnote1 justify=left bold height=14pt "Source Data: SASHELP.CLASS";
footnote2 justify=left bold height=14pt "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold height=14pt "Run on: &RunDayDateTime";
proc print data=sashelp.class noobs
  /* force Arial instead of Calibri,
     control fontsize */
  style(header)=[font=(Arial) fontsize=13pt fontweight=Bold]
  style(data)  =[font=(Arial) fontsize=13pt fontweight=Bold];
where name =: 'J';
run;

title; footnote;
ods excel close; ods results=on;

```


Demo32 Table Over Bar Chart in the Same WorkSheet



```

%let N = 32; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime(RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 105;

ods results=off; ods _all_ close;

ods excel file= "&Path\Demo32 Table Over Bar Chart in the Same
WorkSheet.xlsx"
  nogfootnote /* display the footnotes below the graph image */
  options( embedded_titles='yes' embedded_footnotes='yes'
    title_footnote_nobreak='yes'
    zoom="&ZoomPct"
    sheet_interval='none'
    sheet_name='Table Over Bar Chart' );
title1 height=15pt font='Arial/Bold' justify=left "Demo&N - "
color=red height=14pt
"Table and Bar Chart Stacked "
color=black height=14pt
"- &ZoomPct.% Zoom";
title2 ' ';
title3 font='Arial/Bold' height=15pt justify=left
"All Information about the Students Whose Names Start with 'J'";
footnote;
proc print data=sashelp.class noobs
  /* force Arial instead of Calibri,
  control fontsize */
  style(header)=[font=(Arial) fontsize=13pt fontweight=Bold]
  style(data)  =[font=(Arial) fontsize=13pt fontweight=Bold];

```

```

where name =: 'J';
run;

ods graphics / reset=all scale=off width=6.0in height=3.5in;
title1 height=14pt font='Arial/Bold' "Ranked Height of Students Whose Names
Start with 'J'";
footnote1 justify=left bold height=14pt "Source Data: SASHELP.CLASS";
footnote2 justify=left bold height=14pt "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold height=14pt "Run on: &RunDayDateTime";
proc sgplot data=sashelp.class noborder;
where name =: 'J';
vbar name /
    response=height
    categoryorder=respdesc
    datalabel
    datalabelattrs=(family='Arial' size=12pt weight=Bold)
    fillattrs=(color=CX009900)
    nooutline
    displaybaseline=off
    barwidth=0.4;
yaxis display=none;
xaxis display=(noticks noline nolabel)
    valueattrs=(family='Arial' size=12pt weight=Bold)
    fitpolicy=stagger;
run;

title; footnote;
ods excel close; ods results=on;

```

**Demo33 Hyperlinks and Non-Graphic Images Over and Under Table
(cropped and enlarged for better viewability)**

A1 *f*

	A	B	C	D	E	F	G	H	I
1									
2									
3									
4									
5									
6									
7									
8									
9									
10									
11	Demo33 - HyperLinks and Images Over Table Over Image - 100% Zoom								
12	Above: James H. Goodnight, one of the SAS Institute founders and its CEO								
13	For more information see Wikipedia (also source of photo)								
14									
15	Name	Sex							
16	James	M							
17	Philip	M							
18									
19	Source Data: SASHELP.CLASS								
20	Code: C:\MWSUG_2019\ODS_Excel\code\Demo33.sas								
21	Run on: Wednesday, 31 July 2019 at 9:00:07 PM								
22	Below: Philip R. Holland (center of left photo section) and Phil Mason (in profile at right)								
23	SAS experts / SAS authors at 2017 SAS UK Professional Conference								
24	(Photo by Tricia Aanderud)								
25	See LinkedIn for more information about Phil Holland								
26	See LinkedIn for more information about Phil Mason								
27									
28									
29									
30									
31									
32									
33									
34									
35									
36									
37									
38									
39									
40									
41									
42									
43									
44									

Image Over Table Over Image

Ready

/* The image files used here would be too large to permit the composite result to fit in the default vertical space of the worksheet. So height and width controls are used to shrink them. */

/* The insert_image feature of Excel_Enhance CANNOT resize an image. */

```
%let N = 33; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 100;
```

```

ods results=off; ods _all_ close;
ods excel file= "&Path\Demo33 Hyperlinks and Images Over and Under
Table.xlsx"
  options( embedded_titles='yes' embedded_footnotes='yes'
           title_footnote_nobreak='yes'
           zoom="&ZoomPct"
           sheet_interval='none'
           sheet_name='Image Over Table Over Image' );

data _null_;
  dcl odsout obj();
  * obj.image(file: "&ImgPath.\James_Goodnight.jpg");
  obj.image(file: "&ImgPath.\James_Goodnight.jpg",
           width: "1.37in", /* width should be 1.455in, but then the delivered result
is wider than the original proportions */
           height: "1.39in");
run;

title1 justify=left bold height=12pt
       "Demo&N - "
       color=red height=13pt
       "HyperLinks and Images Over Table Over Image "
       color=black height=12pt
       "- &ZoomPct.% Zoom";

title2
  font='Arial/Bold' height=12pt justify=left color=blue "Above: "
  height=12pt color=black "James H. Goodnight, one of the SAS Institute
founders and its CEO";
title3 link="https://en.wikipedia.org/wiki/James_Goodnight"
  font='Arial/Bold' height=12pt justify=left color=blue underlin=1
  "For more information see Wikipedia (also source of photo)";
footnote1 justify=left bold height=12pt "Source Data: SASHELP.CLASS";
footnote2 justify=left bold height=12pt "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold height=12pt "Run on: &RunDayDateTime";
footnote4 justify=left bold height=12pt color=blue "Below: "
  height=12pt color=black "Philip R. Holland (center of left photo section)
and Phil Mason (in profile at right)";
footnote5 justify=left bold height=12pt "SAS experts / SAS authors at 2017
SAS UK Professional Conference";
footnote6 justify=left bold height=12pt "(Photo by Tricia Aanderud)";
footnote7 link="https://www.linkedin.com/in/hollandnumerics/"
  font='Arial/Bold' height=12pt justify=left color=blue underlin=1
  "See LinkedIn for more information about Phil Holland";
footnote8 link="https://www.linkedin.com/in/philmason/?originalSubdomain=uk"
  font='Arial/Bold' height=12pt justify=left color=blue underlin=1
  "See LinkedIn for more information about Phil Mason";
proc print data=sashelp.class noobs
  /* force Arial instead of Calibri,
  control fontsize */
  style(header)=[font=(Arial) fontsize=12pt fontweight=Bold]
  style(data) =[font=(Arial) fontsize=12pt fontweight=Bold];
where Name IN ('James' 'Philip');
var Name Sex;

```


```
run;

data _null_;
  dcl odsout obj();
  * obj.image(file: "&ImgPath.\Phil_and_Phil.png");
  obj.image(file: "&ImgPath.\Phil_and_Phil.png",
    width: "6in",
    height: "4.5in");
run;

title; footnote;
ods excel close; ods results on;
```

The image insertion code in this example uses the SAS Report Writing Interface. To learn more about using the SAS Report Writing Interface for ODS Excel, see Chevell Parker's presentation and paper "Power Up Your Reporting Using the SAS Output Delivery System".

Demo34 ODS Excel Index
(three worksheets cropped to fit side by side)




```

/* Opens on ODS Excel Index - Click Girls or Boys tab. */
%let N = 34; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 240;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
title1 justify=left bold
    "Demo&N - "
    color=red
    "WorkBook With an ODS Excel Index "
    color=black
    "- &ZoomPct.% Zoom";
ods results off; ods _all_ close;
ods excel file="&Path\Demo34 ODS Excel Index.xlsx"
    options( embedded_titles='yes' embedded_footnotes='yes'
        title_footnote_nobreak='yes'
        zoom="&ZoomPct" sheet_interval='proc' index='yes' );
ods excel options( sheet_name='Girls' );
title2 justify=left bold
    "Female Students - &ZoomPct.% Zoom";
proc print data=sashelp.class noobs; where sex eq 'F'; run;
ods excel options( sheet_name='Boys' );
title2 justify=left bold
    "Male Students - &ZoomPct.% Zoom";
proc print data=sashelp.class noobs; where sex eq 'M'; run;
title; footnote;
ods excel close; ods results on;

```

**Demo35 ODS Excel Index With All WorkSheets InterLinked
(after clicking on Girls on the Index worksheet)**



```

%let N = 35; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 210;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";


title1 justify=left bold
      "Demo&N - "
      color=red
      "Using ODS Excel Index and All WorkSheets InterLinked"
      color=black
      " - &ZoomPct.% Zoom";
ods results off; ods _all_ close;
ods excel
      file="&Path\Demo35 ODS Excel Index With All WorkSheets InterLinked.xlsx"
      options( embedded_titles='yes' embedded_footnotes='yes'
              title_footnote_nobreak='yes'
              sheet_interval='proc' index='yes' zoom="&ZoomPct" );
ods excel options(sheet_name='Girls');
title2 justify=left bold "Female Students";
title3 justify=left bold color=blue underlin=1
      link="#"Index"!a1" "Return to the Index";
title4 justify=left bold color=blue underlin=1
      link="#"Boys"!a1" "Information for Male Students";
proc print data=sashelp.class noobs;
where sex eq 'F'; run;

```




```
ods excel options(sheet_name='Boys');
title2 justify=left bold "Male Students";
title3 justify=left bold color=blue underlin=1
      link="#"Index"!a1" "Return to the Index";
title4 justify=left bold color=blue underlin=1
      link="#"Girls"!a1" "Information for Female Students";
proc print data=sashelp.class noobs;
where sex eq 'M'; run;
title; footnote;
ods excel close; ods results on;
```

Demo36 ODS Excel Index with SAS ByGroup Processing



After Clicking on Product – Slipper:



/* NOTE: Providing a Totals Row for Every BY Group */

```

%let N = 36; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";
%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 215;
  
```

```

footnote1 justify=left bold "Source Data: SASHELP.SHOES";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";

proc summary data=sashelp.shoes;
class Product Region;
var sales;
output out=work.SalesSummary(drop=_freq_ where=( _type_ IN ( 2 , 3) ) sum=;
run;

data work.ToPrint(drop=_type_);
set work.SalesSummary;
if _type_ EQ 2 then Region = '- All Regions Total';
run;

proc sort data=ToPrint;
by Product Region;
run;


ods results off; ods _all_ close;
ods excel
  file="&Path\Demo36 ODS Excel Index with SAS ByGroup Processing.xlsx"
  options(embedded_titles='yes' embedded_footnotes='yes'
    title_footnote_nobreak='yes'
    sheet_interval='bygroup' suppress_bylines='yes'
    sheet_label='Product' /* sheet_name prefix */
    zoom="&ZoomPct" index='yes' );

title1 justify=left bold
"Demo&N - " 'Shoe Sales By Region for #byval(Product) ' "- &ZoomPct.% Zoom";
title2 justify=left bold color=blue
  color=blue
'Using TITLE1 . . . #byval (Product) with PROC PRINT; BY Product; PAGEBY
Product;';
title3 justify=left bold color=blue
"ODS Excel options( suppress_bylines='yes' sheet_interval='bygroup'
index='yes' )";
proc print data=work.ToPrint noobs;
by Product;
pageby Product;
run;

title; footnote;
ods excel close; ods results on;


```

**Putting it all together:
 Demo37 ByGroup Processing With Excel Index and LinkBack
 (after selecting Product – Slipper on the Index worksheet)**




```
/* In the code previously used for Demo36,
  replace TITLE2 and TITLE3 with this statement: */
title2 justify=left bold color=blue underlin=1
link="#'Index'!a1" "Return to the Index";
```

**Demo38 ODS Excel Table of Contents With All WorkSheets InterLinked
(each PROC output has two links)**



After selecting a link to Boys information:



/* NOTE: if there were more worksheets, interlinking them would be more coding work. Only the links FROM the index are "automatic". */

```

%let N = 38; /* Demo Step */
%let Path      = C:\MWSUG_2019\ODS_Excel\results;
%let CodePath  = C:\MWSUG_2019\ODS_Excel\code;
%let ImgPath   = C:\MWSUG_2019\ODS_Excel\images;
%include "&CodePath.\macros\RunDayDateTime.sas";

```

```

%RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO);
%let ZoomPct = 210;
footnote1 justify=left bold "Source Data: SASHELP.CLASS";
footnote2 justify=left bold "Code: &CodePath\Demo&N..sas";
footnote3 justify=left bold "Run on: &RunDayDateTime";
title1 justify=left bold
    "Demo&N - "
    color=red
    "Using ODS Excel Table of Contents & InterLinked WorkSheets"
    color=black
    " - &ZoomPct.% Zoom";
ods results off; ods _all_ close;
ods excel file="&Path\Demo38 ODS Excel Table of Contents With All WorkSheets
InterLinked.xlsx"
    options(embedded_titles='yes' embedded_footnotes='yes'
        title_footnote_nobreak='yes'
        sheet_interval='proc' contents='yes' zoom="&ZoomPct");
ods excel options(sheet_name='Girls');
ods proclabel='Click here for Ages, Heights, & Weights';
title2 justify=left bold "Female Students";
title3 justify=left bold color=blue underlin=1
    link="#'The Table of Contents'!a1" "Return to the Table of Contents";
title4 justify=left bold color=blue underlin=1
    link="#'Boys'!a1" "Information for Male Students";
proc print data=sashelp.class noobs
    contents='of Girls in the Class';
where sex eq 'F'; run;
ods excel options(sheet_name='Boys');
ods proclabel='Click here for Ages, Heights, & Weights';
title2 justify=left bold "Male Students";
title3 justify=left bold color=blue underlin=1
    link="#'The Table of Contents'!a1" "Return to the Table of Contents";
title4 justify=left bold color=blue underlin=1
    link="#'Girls'!a1" "Information for Female Students";
proc print data=sashelp.class noobs
    contents='of Boys in the Class';
where sex eq 'M'; run;
title; footnote;
ods excel close; ods results on;

```

For More About ODS Excel

To learn about ODS Excel features not covered in this paper, please see Chevell Parker's presentation and paper "Power Up Your Reporting Using the SAS Output Delivery System". The ODS Excel topics covered there include:

- Enhancing your Microsoft Excel Worksheets with Formats and Formulas
- SAS Formatting
- Custom Excel Formatting
- Enhancing Worksheets by Using the SAS Report Writing Interface with Formats and Formulas

Conclusion

ODS Excel, supplemented by the Excel_Enhance macro, is the best of the SAS non-DDE tools to create highly formatted reports that can be opened in Excel. For me, its only significant limitation is lack of support for pivot tables. ODS Excel options related to printing have not been presented here. For more information, see the Output Delivery System User's Guide, for which the link to the latest version as of this date is:

<https://documentation.sas.com/?docsetId=odsug&docsetTarget=p09n5pw9ol0897n1qe04zeur27rv.htm&docsetVersion=9.4&locale=en>

References

1. Bessler, LeRoy (2012), "Give Them Exactly What They Want with SAS-to-Excel Via Automation with Dynamic Data Exchange (DDE)", Proceedings of the Summer 2012 Wisconsin Illinois SAS Users Conference, Milwaukee, WI, USA: Software User Services, Inc. Find it at www.wiilsu.org
2. Eberhardt, Peter and Kong, Louanna (2012), "The Armchair Quarterback: Writing SAS Code for the Perfect Pivot (Table, That Is)", Proceedings of SAS Global Forum 2012, Cary, NC, USA: SAS Institute Inc. See <http://support.sas.com/resources/papers/proceedings12/146-2012.pdf>
3. Bessler, LeRoy (2009), "Using SAS to Manage, Monitor, and Control the SAS BI Server: User-Developed Custom Tools for the SAS Server Administrator, User, or Manager", Proceedings of SAS Global Forum 2009, Cary, NC, USA: SAS Institute Inc. See <http://support.sas.com/resources/papers/proceedings09/274-2009.pdf>

Author Information

Your questions, comments, and suggestions are always welcome.

LeRoy Bessler PhD
Mequon, Wisconsin, USA
Visual Data Insights™
Strong Smart Systems™
Le_Roy_Bessler@wi.rr.com

Dr. LeRoy Bessler has presented at conferences in the USA, Canada, and Europe, on effective visual communication (using graphs, tables, web pages, maps, or color), SAS to Excel, tools for SAS server administrators, users, and managers, and Application Development for Reliability, Reusability, Maintainability, Extendibility, and Flexibility. His experience includes application development and supporting users, servers, software, and data. He is writing a book on communication-effective data visualization, a field that he has plied since 1981, and has declared himself to be a Data Artist.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

Visual Data Insights and Strong Smart Systems are trademarks of LeRoy Bessler PhD.

Appendices

Appendix A. RunDayDateTime Macro

Identifying exactly when a report was created can be regarded as good practice. However, if you omit FOOTNOTE3 from these examples, this macro need not be run.

This macro was used in all of the demonstration examples, but is NOT inherent to using ODS Excel. If you wish to include something like FOOTNOTE3 in any application output, this macro delivers the name of the day of the week, the date, and the time when the macro was launched.

If multiple outputs, each with their own invocation of this macro, are created during a SAS session or SAS batch job, each will have a different time stamp.

Unlike macro variable RunDayDateTime created by this macro, the automatic SAS macro variables SYSTIME, SYSDATE, SYSDATE9, and SYSDAY are established only at the start of the SAS session or SAS job.

```
%macro RunDayDateTime (RunDayDateTimeAsFileNameSuffix=NO) ;

%GLOBAL RunDayDateTime FileNameDTsuffix;
DATA _NULL_ ;
RunDate = DATE() ; RunTime = TIME() ;
RunDayDateTimeText =
TRIM(LEFT (PUT (RunDate,weekdatx37.))) || ' at ' ||
TRIM(LEFT (PUT (RunTime,timeampm11.))) ;
CALL SYMPUT ('RunDayDateTime',TRIM(LEFT (RunDayDateTimeText))) ;
RunDayDateTimeFileNameSuffix =
TRIM(LEFT (PUT (RunDate,downname3.))) || '_' ||
TRIM(LEFT (PUT (RunDate,date9.))) || '_' ||
TRIM(LEFT (COMPRESS (PUT (RunTime,TOD8.),':'))) || '_' ||
SUBSTR (PUT (RunTime,TOD12.3),10,3) ; /* prevent duplicate
timestamps when two successive macro invocations run
during the same second */
CALL SYMPUT ('FileNameDTsuffix',
TRIM(LEFT (RunDayDateTimeFileNameSuffix))) ;
RUN ;

%if %upcase (&RunDayDateTimeAsFileNameSuffix) EQ YES
%then %let FileNameDTsuffix = %str (&FileNameDTsuffix) ;
%else %let FileNameDTsuffix = %str () ;

%mend RunDayDateTime ;
```

Appendix B. Excel Enhance Macro & Two Macros Used With It

Excel Enhance Macro

```
/* options noxwait noxsync ;
   This must be used with the macro. */
```

Get the most current version of the macro via download from:

ftp://ftp.sas.com/techsup/download/blind/enhanced_macro.zip

The version of the macro used for this paper was downloaded on 17 August 2019.

If you have any problems with the link above (e.g., if it no longer exists), please contact SAS Technical Support at 1.919.677.8008 or Support@sas.com

CreateTableAsImageFile Macro

```
%macro CreateTableAsImageFile(  
    imagefilenameprefix=,  
    data=,  
    title=,  
    width_inches=,  
    height_inches=,  
    fontsize=12pt);  
  
options papersize=(&width_inches.in &height_inches.in);  
  
ods printer file="c:\temp\&imagefilenameprefix.&data..png" printer=PNG300;  
  
title justify=left font='Arial/Bold' height=&fontsize "&title";  
proc print data=&data noobs  
    style(header) = [fontfamily='Arial' fontsize=&fontsize fontweight=Bold]  
    style(data)    = [fontfamily='Arial' fontsize=&fontsize];  
run;  
  
ods printer close;  
  
%mend CreateTableAsImageFile;
```

BlankRowInWorkSheet Macro

```
%macro BlankRowInWorkSheet (Count=,Font=,FontSize=,FontWeight=);  
  
/* Purpose: The Excel_Enhance macro is used to place graphic images  
or tables created as image files into the empty space of an Excel worksheet.  
This macro can be used to create empty rows between title rows at the top  
and footnotes at the bottom of an Excel worksheet to be loaded. */  
  
%do i = 1 %to &Count %by 1;  
p ' '  
    / style={just=left font=(&Font) fontsize=&FontSize fontweight=&FontWeight};  
%end;  
  
%mend BlankRowInWorkSheet;
```

Appendix C. Not Specific to ODS Excel

Many of the examples create a title line in multiple colors.

Though this phenomenon is unique to the ODS Excel destination, I need to point out an unexpected result that involves the SAS TITLE statement (and presumably FOOTNOTE statement as well—I did not test the FOOTNOTE statement for this situation).

In a TITLE (or FOOTNOTE) statement you can change the color of text in the middle of the string, as in, e.g.,

```
TITLE1 "default black part" COLOR=RED "red part" COLOR=BLUE "blue part";
```

The above statement will cause no problems, but if you want to use a non-default text height, as in, e.g.,

```
TITLE1 HEIGHT=16pt "default black part"  
      COLOR=RED "red part"  
      COLOR=BLUE "blue part";
```

all three parts will NOT be drawn with that height in two situations: (a) ANY tabular output; and (b) graphic output created with option NOGTITLE. Using option GTITLE for the graphic output DOES use the specified height for ALL parts of the TITLE. (With option GTITLE, any TITLE lines are drawn inside the graphic image file.)

For tabular PROC output and for NOGTITLE graphic output, it is necessary to use, e.g.,

```
TITLE1 HEIGHT=16pt "default black part"  
      HEIGHT=16pt COLOR=RED "red part"  
      HEIGHT=16pt COLOR=BLUE "blue part";
```

See Demo27 and Demo30 for examples of application of this tip.